

The *Countdown* problem

Introduction

Countdown is the name of a game from a popular British television programme; in France it is called *Le Conte est Bon*. Contestants are given six source numbers, not necessarily all different, and a target number, all of which are positive integers. The aim is to use some of the source numbers to build an arithmetic expression whose value is as close to the target as possible. Expressions are constructed using only the four basic operations of addition, subtraction, multiplication and division. Contestants are allowed 30 s thinking time. For example, with source numbers $[1, 3, 7, 10, 25, 50]$ and target 831 there is no exact solution; one expression that comes closest is $7 + (1 + 10) \times (25 + 50) = 832$. Our aim in this pearl is to describe various programs for solving *Countdown*, all based in one way or another on exhaustive search. *Countdown* is attractive as a case study in exhaustive search because the problem is simply stated and the different solutions illustrate the space and time trade-offs that have to be taken into account in comparing functional programs.

A simple program

Here is a straightforward program for *Countdown*:

```
countdown1    :: Int → [Int] → (Expr, Value)
countdown1 n  = nearest n · concatMap mkExprs · subseqs
```

First of all, the source numbers are given as a list; the order of the elements is unimportant, but duplicates do matter. We will suppose that the list is in ascending order, a fact that is exploited later on. Each selection is therefore represented by a nonempty subsequence. For each subsequence xs , all possible arithmetic expressions that can be constructed from xs are

determined, along with their values.¹ The results are concatenated and one nearest the target is selected.

The ingredients making up *countdown1* are defined as follows. First, *subseqs* returns a list of all the nonempty subsequences of a nonempty list:

$$\begin{aligned} \text{subseqs } [x] &= [[x]] \\ \text{subseqs } (x : xs) &= xss \mathrel{++} [x] : \text{map } (x :) xss \\ &\quad \textbf{where } xss = \text{subseqs } xs \end{aligned}$$

Next, the data types of expressions and values can be declared by

$$\begin{aligned} \textbf{data Expr} &= \text{Num Int} \mid \text{App Op Expr Expr} \\ \textbf{data Op} &= \text{Add} \mid \text{Sub} \mid \text{Mul} \mid \text{Div} \\ \textbf{type Value} &= \text{Int} \end{aligned}$$

The value of an expression is computed by

$$\begin{aligned} \text{value} &:: \text{Expr} \rightarrow \text{Value} \\ \text{value } (\text{Num } x) &= x \\ \text{value } (\text{App op } e1 \ e2) &= \text{apply op } (\text{value } e1) (\text{value } e2) \end{aligned}$$

where *apply Add* = (+), *apply Sub* = (−) and so on. However, not all possible expressions are valid in *Countdown*. For instance, the result of a subtraction should be a positive integer, and division is valid only when the divisor divides the dividend exactly. An expression is valid if its subexpressions are, and if the operation at the root passes the test *legal*, where

$$\begin{aligned} \text{legal} &:: \text{Op} \rightarrow \text{Value} \rightarrow \text{Value} \rightarrow \text{Bool} \\ \text{legal Add } v1 \ v2 &= \text{True} \\ \text{legal Sub } v1 \ v2 &= (v2 < v1) \\ \text{legal Mul } v1 \ v2 &= \text{True} \\ \text{legal Div } v1 \ v2 &= (v1 \bmod v2 == 0) \end{aligned}$$

The next ingredient is *mkExpr*, which creates a list of all legal expressions that can be built using the given subsequence:

$$\begin{aligned} \text{mkExprs} &:: [\text{Int}] \rightarrow [(\text{Expr}, \text{Value})] \\ \text{mkExprs } [x] &= [(\text{Num } x, x)] \\ \text{mkExprs } xs &= [\text{ev} \mid (ys, zs) \leftarrow \text{unmerges } xs, \\ &\quad \text{ev1} \leftarrow \text{mkExprs } ys, \\ &\quad \text{ev2} \leftarrow \text{mkExprs } zs, \\ &\quad \text{ev} \leftarrow \text{combine ev1 ev2}] \end{aligned}$$

¹ Logically there is no need to return both expressions and values as the latter can be determined from the former. But, as we have seen in the pearl “Making a century” (Pearl 6), it is a good idea to avoid computing values more than once, so this optimisation has been incorporated from the outset.

Given an ordered list xs of length greater than one, $unmerges\ xs$ is a list of all pairs (ys, zs) of nonempty lists such that $merge\ ys\ zs = xs$, where $merge$ merges two ordered lists into one (it is in the specification of $unmerges$ that we exploit the fact that inputs are ordered). One way of defining $unmerges$ is as follows:

$$\begin{aligned} unmerges &:: [a] \rightarrow [[a], [a]] \\ unmerges\ [x, y] &= [[x], [y]], ([y], [x]) \\ unmerges\ (x : xs) &= [[x], xs], (xs, [x]) \mathrel{++} \\ &\quad concatMap\ (add\ x)\ (unmerges\ xss) \\ &\quad \textbf{where}\ add\ x\ (ys, zs) = [(x : ys, zs), (ys, x : zs)] \end{aligned}$$

It is an instructive exercise to calculate this definition of $unmerges$ from its specification, but we will leave that pleasure to the reader.

The function $combine$ is defined by

$$\begin{aligned} combine &:: (Expr, Value) \rightarrow (Expr, Value) \rightarrow [(Expr, Value)] \\ combine\ (e1, v1)\ (e2, v2) &= [(App\ op\ e1\ e2, apply\ op\ v1\ v2) \mid op \leftarrow ops, legal\ op\ v1\ v2] \end{aligned}$$

where $ops = [Add, Sub, Mul, Div]$.

Finally, the function $nearest\ n$ takes a nonempty list of expressions and returns some expression in the list whose value is nearest the target n . We also want to stop searching the list if and when an expression is found whose value matches the target exactly:

$$\begin{aligned} nearest\ n\ ((e, v) : evs) &= \textbf{if}\ d == 0\ \textbf{then}\ (e, v) \\ &\quad \textbf{else}\ search\ n\ d\ (e, v)\ evs \\ &\quad \textbf{where}\ d = abs\ (n - v) \\ search\ n\ d\ ev\ [] &= ev \\ search\ n\ d\ ev\ ((e, v) : evs) &\mid\ d' == 0 = (e, v) \\ &\mid\ d' < d = search\ n\ d'\ (e, v)\ evs \\ &\mid\ d' \geq d = search\ n\ d\ ev\ evs \\ &\quad \textbf{where}\ d' = abs\ (n - v) \end{aligned}$$

For example, under GHCi (version 6.8.3 running on a 2394MHz laptop under Windows XP) we have

```
> display (countdown1 831 [1,3,7,10,25,50])
(7+((1+10)*(25+50))) = 832
(42.28 secs, 4198816144 bytes)
> length $ concatMap mkExprs $ subseqs [1,3,7,10,25,50]
4672540
```

So *countdown1* takes about 42 s to determine and analyse about 4.5 million expressions, about 100 000 expressions per second. This is not within the 30 s limit, so is not good enough.

Two optimisations

There are two simple optimisations that can help improve matters. The first concerns the legality test. There are about 33 million expressions that can be built from six numbers, of which, depending on the input, between 4 million and 5 million are legal. But there is a great deal of redundancy. For example, each of the following pairs of expressions is essentially the same:

$$x + y \text{ and } y + x, \quad x * y \text{ and } y * x, \quad (x - y) + z \text{ and } (x + z) - y$$

A stronger legality test is provided by

$$\begin{aligned} \text{legal Add } v1 \ v2 &= (v1 \leq v2) \\ \text{legal Sub } v1 \ v2 &= (v2 < v1) \\ \text{legal Mul } v1 \ v2 &= (1 < v1) \wedge (v1 \leq v2) \\ \text{legal Div } v1 \ v2 &= (1 < v2) \wedge (v1 \bmod v2 = 0) \end{aligned}$$

This stronger test takes account of the commutativity of $+$ and $*$ by requiring that arguments be in numerical order, and the identity properties of $*$ and $/$ by requiring that their arguments be non-unitary. This test reduces the number of legal expressions to about 300 000. One can go further and strengthen the legality test yet more, but we will leave that to the next section.

The second optimisation concerns *unmerges* and *combine*. As defined above, *unmerges xs* returns all pairs (ys, zs) such that $\text{merge } ys \ zs = xs$, and that means each pair is generated twice, once in the form (ys, zs) and once in the form (zs, ys) . There is no need to double the work, and we can redefine *unmerges* so that it returns only the essentially distinct pairs:

$$\begin{aligned} \text{unmerges } [x, y] &= [[x], [y]] \\ \text{unmerges } (x : xs) &= [[x], xs] \text{ ++ concatMap (add } x) (\text{unmerges } xs) \\ &\quad \textbf{where } \text{add } x \ (ys, zs) = [(x : ys, zs), (ys, x : zs)] \end{aligned}$$

The function *combine* can be easily modified to take account of the new *unmerges*:

$$\begin{aligned} \text{combine } (e1, v1) \ (e2, v2) \\ = \ [(\text{App } op \ e1 \ e2, \text{apply } op \ v1 \ v2) \mid op \leftarrow ops, \text{legal } op \ v1 \ v2] \text{ ++} \\ \ [(\text{App } op \ e2 \ e1, \text{apply } op \ v2 \ v1) \mid op \leftarrow ops, \text{legal } op \ v2 \ v1] \end{aligned}$$

```

comb1 (e1, v1) (e2, v2)
  = [(App Add e1 e2, v1 + v2), (App Sub e2 e1, v2 - v1)] ++
    if 1 < v1 then
      [(App Mul e1 e2, v1 * v2)] ++ [(App Div e2 e1, q) | r = 0]
    else []
    where (q, r) = divMod v2 v1
comb2 (e1, v1) (e2, v2)
  = [(App Add e1 e2, v1 + v2)] ++
    if 1 < v1 then
      [(App Mul e1 e2, v1 * v2), (App Div e1 e2, 1)]
    else []

```

Fig. 20.1 Definitions of *comb1* and *comb2*

However, a faster method is to incorporate the stronger legality test directly into the definition of *combine*:

```

combine (e1, v1) (e2, v2)
  | v1 < v2  = comb1 (e1, v1) (e2, v2)
  | v1 == v2 = comb2 (e1, v1) (e2, v2)
  | v1 > v2  = comb1 (e2, v2) (e1, v1)

```

The function *comb1* is used when the first expression has a value strictly less than the second, and *comb2* when the two values are equal. Their definitions are given in Figure 20.1. Installing these changes leads to *countdown2*, whose definition is otherwise the same as *countdown1*. For example:

```

> display (countdown2 831 [1,3,7,10,25,50])
(7+((1+10)*(25+50))) = 832
(1.77 secs, 168447772 bytes)
> length $ concatMap mkExprs $ subseqs [1,3,7,10,25,50]
240436

```

This is better, in that it takes only about 2s to determine and analyse about 250 000 expressions, but there is still room for improvement.

An even stronger legality test

In an attempt to restrict still further the number of expressions that have to be considered, let us say that an expression is in *normal form* if it is a sum of the form

$$[(e_1 + e_2) + \cdots + e_m] - [(f_1 + f_2) + \cdots + f_n]$$

where $m \geq 1$ and $n \geq 0$, both e_1, e_2, \dots and f_1, f_2, \dots are in ascending order of value, and each e_j and f_j is a product of the form

$$[(g_1 * g_2) * \dots * g_p] / [(h_1 * h_2) * \dots * h_q]$$

where $p \geq 1$ and $q \geq 0$, both g_1, g_2, \dots and h_1, h_2, \dots are in ascending order of value and each g_j and h_j is either a single number or an expression in normal form.

Up to rearrangements of subexpressions with equal values, each expression has a unique normal form. Of the 300 000 expressions over six numbers that are legal according to the earlier definition, only about 30 000 to 70 000 are in normal form. However, normal form does not eliminate redundancy completely. For example, the expressions $2 + 5 + 7$ and $2 * 7$ have the same value, but the latter is built out of numbers that are a subsequence of the former. There is, therefore, no need to build the former. But we will not explore the additional optimisation of “thinning” a list of expressions to retain only the really essential ones. Experiments show that thinning turns out to be not worth the candle: the savings made in analysing only the really essential expressions are outweighed by the amount of effort needed to determine them.

We can capture normal forms by strengthening the legality test, but this time we have to consider expressions as well as values. First let us define *non* by

$$\begin{aligned} \text{non} &:: Op \rightarrow Expr \rightarrow Bool \\ \text{non } op \ (Num \ x) &= True \\ \text{non } op1 \ (App \ op2 \ e1 \ e2) &= op1 \neq op2 \end{aligned}$$

Then the stronger legality test is implemented by

$$\begin{aligned} \text{legal} &:: Op \rightarrow (Expr, Value) \rightarrow (Expr, Value) \rightarrow Bool \\ \text{legal } Add \ (e1, v1) \ (e2, v2) &= (v1 \leq v2) \wedge \text{non } Sub \ e1 \wedge \text{non } Add \ e2 \wedge \text{non } Sub \ e2 \\ \text{legal } Sub \ (e1, v1) \ (e2, v2) &= (v2 < v1) \wedge \text{non } Sub \ e1 \wedge \text{non } Sub \ e2 \\ \text{legal } Mul \ (e1, v1) \ (e2, v2) &= (1 < v1 \wedge v1 \leq v2) \wedge \text{non } Div \ e1 \wedge \text{non } Mul \ e2 \wedge \text{non } Div \ e2 \\ \text{legal } Div \ (e1, v1) \ (e2, v2) &= (1 < v2 \wedge v1 \bmod v2 = 0) \wedge \text{non } Div \ e1 \wedge \text{non } Div \ e2 \end{aligned}$$

Just as before, we can incorporate the above legality test into a modified definition of *combine*. It is necessary only to change *comb1* and *comb2*. The revised definitions are given in [Figure 20.2](#).

```

comb1 (e1, v1) (e2, v2)
= (if non Sub e1 ∧ non Sub e2 then
  [(App Add e1 e2, v1 + v2) | non Add e2] ++ [(App Sub e2 e1, v2 - v1)]
  else []) ++
  (if 1 < v1 ∧ non Div e1 ∧ non Div e2 then
    [(App Mul e1 e2, v1 * v2) | non Mul e2] ++ [(App Div e2 e1, q) | r == 0]
    else [])
  where (q, r) = divMod v2 v1
comb2 (e1, v1) (e2, v2)
= [(App Add e1 e2, v1 + v2) | non Sub e1, non Add e2, non Sub e2] ++
  (if 1 < v1 ∧ non Div e1 ∧ non Div e2 then
    [(App Mul e1 e2, v1 * v2) | non Mul e2] ++ [(App Div e1 e2, 1)]
    else [])

```

Fig. 20.2 New definitions of *comb1* and *comb2*

Calling the result of installing these changes *countdown3*, we have

```

> display (countdown3 831 [1,3,7,10,25,50])
(7+((1+10)*(25+50))) = 832
(1.06 secs, 88697284 bytes)
> length $ concatMap mkExprs $ subsequences [1,3,7,10,25,50]
36539

```

Now it takes only 1s to determine and analyse about 36 000 expressions, which is roughly double the speed of *countdown2*.

Memoisation

Even ignoring the redundancy in the set of expressions being determined, computations are repeated because every subsequence is treated as an independent problem. For instance, given the source numbers [1 .. 6], expressions with basis [1 .. 5] will be computed twice, once for the subsequence [1 .. 5] and once for [1 .. 6]. Expressions with basis [1 .. 4] will be computed four times, once for each of the subsequences

[1, 2, 3, 4], [1, 2, 3, 4, 5], [1, 2, 3, 4, 6], [1, 2, 3, 4, 5, 6]

In fact, expressions with a basis of k numbers out of n source numbers will be computed 2^{n-k} times.

One way to avoid repeated computations is to memoise the computation of *mkExprs*. In memoisation, the top-down structure of *mkExprs* is preserved but computed results are remembered and stored in a memo table

for subsequent retrieval. To implement memoisation we need a data type *Memo* on which the following operations are supported:

```
empty  :: Memo
fetch  :: Memo → [Int] → [(Expr, Value)]
store  :: [Int] → [(Expr, Value)] → Memo → Memo
```

The value *empty* defines an empty memo table, *fetch* takes a list of source numbers and looks up the computed expressions for the list, while *store* takes a similar list together with the expressions that can be built from them, and stores the result in the memo table.

We can now rewrite *mkExprs* to read

```
mkExprs          :: Memo → [Int] → [(Expr, Value)]
mkExprs memo [x] = [(Num x, x)]
mkExprs memo xs  = [ev | (ys, zs) ← unmerges xs,
                        ev1 ← fetch memo ys,
                        ev2 ← fetch memo zs,
                        ev ← combine ev1 ev2]
```

This code assumes that for any given subsequence *xs* of the input, all the arithmetic expressions for *ys* and *zs* for each possible split of *xs* have already been computed and stored in the memo table. This assumption is valid if we list and process the subsequences of the source numbers in such a way that if *xs* and *ys* are both subsequences of these numbers, and *xs* is a subsequence of *ys*, then *xs* appears before *ys* in the list of subsequences. Fortunately, the given definition of *subseqs* does possess exactly this property. We can now define

```
countdown4      :: Int → [Int] → (Expr, Value)
countdown4 n    = nearest n · extract · memoise · subseqs
```

where *memoise* is defined by

```
memoise          :: [[Int]] → Memo
memoise          = foldl insert empty
insert memo xs    = store xs (mkExprs memo xs) memo
```

The function *extract* flattens a memo table, returning a list of all the expressions in it. This function is defined below when we fix on the structure of *Memo*.

One possible structure for *Memo* is a trie:

```
data Trie a      = Node a [(Int, Trie a)]
type Memo        = Trie [(Expr, Value)]
```


A trie is a Rose tree whose branches are labelled, in this case with an integer. The empty memo table is defined by $empty = Node [] []$. We search a memo table by following the labels on the branches:

$$\begin{aligned}
 fetch & :: Memo \rightarrow [Int] \rightarrow [(Expr, Value)] \\
 fetch (Node es xms) [] & = es \\
 fetch (Node es xms) (x : xs) & = fetch (follow x xms) xs \\
 follow & :: Int \rightarrow [(Int, Memo)] \rightarrow Memo \\
 follow x xms & = head [m \mid (x', m) \leftarrow xms, x == x']
 \end{aligned}$$

Note that searching a table for an entry with label xs returns an undefined result (the head of an empty list) if there is no path in the trie whose branches are labelled with xs . But this is not a problem, because the definition of *subseqs* guarantees that entries are computed in the right order, so all necessary entries will be present.

Here is how we store new entries:

$$\begin{aligned}
 store & :: [Int] \rightarrow [(Expr, Value)] \rightarrow Memo \rightarrow Memo \\
 store [x] es (Node fs xms) & = Node fs ((x, Node es []) : xms) \\
 store (x : xs) es (Node fs xms) & \\
 = Node fs (yms \mathbin{++} (x, store xs es m) : zms) & \\
 \textbf{where } (yms, (z, m) : zms) & = break (equals x) xms \\
 equals x (z, m) & = (x == z)
 \end{aligned}$$

The definition of *store* assumes that if an entry for $xs \mathbin{++} [x]$ is new, then the entries for xs are already present in the table. The Haskell function *break p* was defined in the previous pearl.

Finally, we can extract all entries from a memo table by

$$\begin{aligned}
 extract & :: Memo \rightarrow [(Expr, Value)] \\
 extract (Node es xms) & = es \mathbin{++} concatMap (extract \cdot snd) xms
 \end{aligned}$$

Now we have, for example:

```
> display (countdown4 831 [1,3,7,10,25,50])
(10*((1+7)+(3*25))) = 830
(0.66 secs, 55798164 bytes)
```

The computation returns a different expression, owing to the different order in which expressions are analysed, but at a cost of about half that of *countdown3*.

Skeleton trees

Memoisation of *countdown* comes at a cost: building the memo table makes heavy demands on the heap and much time is spent in garbage collection. How can we keep the advantage of memoisation while reducing space requirements?

Suppose we ignore the operators in an expression, focusing only on the parenthesis structure. How many different oriented binary trees can we build? In an oriented tree the order of the subtrees is not taken into account. We exploited this idea in an “oriented” definition of *unmerges*. It turns out that there are only 1881 oriented binary trees with a basis included in six given numbers. An oriented binary tree may also be called a *skeleton* tree. For an algorithm that is economical in its use of space we could, therefore, build these trees first, and only afterwards insert the operators.

Pursuing this idea, consider the following type of tip-labelled binary tree:

data *Tree* = *Tip Int* | *Bin Tree Tree*

Instead of memoising expressions we can memoise trees:

type *Memo* = *Trie* [*Tree*]

We can build trees in exactly the same way as we built expressions:

mkTrees :: *Memo* → [*Int*] → [*Tree*]
mkTrees memo [*x*] = [*Tip x*]
mkTrees memo xs = [*Bin t1 t2* | (*ys, zs*) ← *unmerges xs*,
 t1 ← *fetch memo ys*,
 t2 ← *fetch memo zs*]

We can convert a tree into a list of expressions by inserting operators in all legal ways:

toExprs :: *Tree* → [(*Expr*, *Value*)]
toExprs (*Tip x*) = [(*Num x, x*)]
toExprs (*Bin t1 t2*) = [*ev* | *ev1* ← *toExprs t1*, *ev2* ← *toExprs t2*,
 ev ← *combine ev1 ev2*]

Now we have

countdown5 n
 = *nearest n · concatMap toExprs · extract · memoise · subseqs*

File	<i>countdown1</i>		<i>countdown2</i>		<i>countdown3</i>		<i>countdown4</i>		<i>countdown5</i>	
	Total	GC	Total	GC	Total	GC	Total	GC	Total	GC
d6	1.56	0.78	0.19	0.08	0.09	0.05	0.08	0.02	0.05	0.00
d7	77.6	36.9	2.03	1.19	0.44	0.09	0.53	0.30	0.33	0.02
d8	—	—	99.8	57.2	13.8	7.30	16.9	9.02	7.22	0.31

Fig. 20.3 Running times of *countdown* for inputs of six, seven and eight source numbers

where *memoise* is defined by

```
memoise      :: [[Int]] → Memo
memoise      = foldl insert empty
insert memo xs = store xs (mkTrees memo xs) memo
```

Running our standard example yields

```
> display (countdown5 831 [1,3,7,10,25,50])
(10*((1+7)+(3*25))) = 830
(1.06 secs, 88272332 bytes)
```

So it seems on the evidence of this single test that memoising skeleton trees rather than expressions may not have been such a good idea. But the situation merits a closer look.

A further experiment

Let us see how the five versions of *countdown* described above perform with an optimising compiler. We compiled the five programs under GHC, version 6.8.3, with the *-O2* flag set. The statistics were gathered using GHC’s run-time system with the *-s* flag. There were three files, *d6*, *d7* and *d8* containing six, seven and eight source numbers respectively. In each case we ensured there was no exact match, so the full space of possible expressions was explored. The statistics are provided in Figure 20.3. Shown are the total time and the time spent in garbage collection; all times are in seconds. The program *countdown1* was not run on *d8*.

Three main conclusions can be drawn from the experiment. First and most obviously, compilation gives a substantial improvement over interpretation. Second, for six or seven source numbers, there is not much difference between *countdown3* (the version with the strong legality test), *countdown4* (the version with both the strong legality test and memoisation) and *countdown5* (the version with the strong legality test and memoisation of skeleton trees

rather than expressions). But for eight source numbers, the final version *countdown5* has begun to pull away, running about twice as fast as the others, mostly owing to the reduced time spent in garbage collection, in fact about 5% of the total time, compared with about 50% for *countdown3* and *countdown4*.

Final remarks

This pearl has been based on material extracted and modified from Bird and Mu (2005), which presents the specification of *Countdown* in a relational setting, and goes on to calculate a number of programs using the algebraic laws of fold and unfold. None of these calculations has been recorded above. *Countdown* was first studied in an earlier pearl (Hutton, 2002) as an illustration of how to prove that functional programs meet their specification. Hutton's aim was not to derive the best possible algorithm, but to present one whose correctness proof required only simple induction. Essentially, Hutton's proof dealt with the correctness of *countdown2*.

References

- Bird, R. S. and Mu, S.-C. (2005). *Countdown: a case study in origami programming*. *Journal of Functional Programming* **15** (6), 679–702.
- Hutton, G. (2002). The *Countdown* problem. *Journal of Functional Programming* **12** (6), 609–16.